

Techniques for Calculations in Quantum Field Theory: W-strahlung in e^-e^\pm colliders

J. C. Romão*

*Departamento de Física and CFTP, Instituto Superior Técnico
Avenida Rovisco Pais 1, 1049-001 Lisboa, Portugal*

In this note we review the techniques used in Quantum Field Theory in a non trivial example, the scatterings $e^- + e^- \rightarrow W^- + e^- + \nu_e$, $e^- + e^+ \rightarrow W^- + e^+ + \nu_e$ and $e^- + e^+ \rightarrow W^+ + e^- + \bar{\nu}_e$.

I. INTRODUCTION

It is always a problem when the students want to attack a more complex problem that needs large trace calculations and numerical integrations. In normal textbooks this part it is normally not covered or, at most, is left as an exercise. My experience is that, for the first time, the students need some help. I will try to cover in this article all the details of a real calculation, from the first stages to the final plots for the cross section. To this end we pick a non-trivial problem, the W -strahlung in e^-e^\pm collisions. More precisely we consider the processes,

$$e^-(p_1) + e^-(p_2) \rightarrow e^-(p_3) + W^-(p_4) + \nu_e(p_5) \quad (1)$$

$$e^-(p_1) + e^+(p_2) \rightarrow e^+(p_3) + W^-(p_4) + \nu_e(p_5) \quad (2)$$

$$e^-(p_1) + e^+(p_2) \rightarrow e^-(p_3) + W^+(p_4) + \bar{\nu}_e(p_5) , \quad (3)$$

in the limit of neglecting the electron mass.

II. $e^-e^- \rightarrow W^-e^-\nu_e$

A. Diagrams

This process corresponds to the Feynman diagrams of Fig. 1. Note that there is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.

B. The Amplitudes

We start by noting that the amplitudes can be written in the following form:

$$M_i = \sum_{i=1}^6 \bar{u}(p_5)\Gamma_1^{A_i}u(p_1)\bar{u}(p_3)\Gamma_2^{A_i}u(p_2)\frac{1}{D_{A_i}} - \sum_{i=1}^6 \bar{u}(p_3)\Gamma_1^{B_i}u(p_1)\bar{u}(p_5)\Gamma_2^{B_i}u(p_2)\frac{1}{D_{B_i}} \quad (4)$$

where we have put explicitly the minus sign between the two sets of diagrams, which we denote by A and B . We have

$$\Gamma_1^{A1} = \frac{eg}{\sqrt{2}}\gamma^\mu P_L(\not{p}_4 + \not{p}_5)\gamma^{\alpha_1} \quad \Gamma_2^{A1} = e \gamma_{\alpha_1}$$

*Electronic address: jorge.romao@ist.utl.pt




Figure 1: Feynman diagrams for $e^-e^- \rightarrow e^-W^-\nu_e$. There is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.

$$\begin{aligned}
\Gamma_1^{A2} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma^{\alpha_2} (g_- P_L + g_+ P_R) & \Gamma_2^{A2} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_2} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A3} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^{\alpha_3} P_L (\not{p}_1 - \not{p}_4) \gamma^\mu P_L & \Gamma_2^{A3} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_3} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A4} &= \frac{g}{\sqrt{2}} \gamma_{\rho_1} P_L \Gamma_{\text{WWA}}^{\rho_1 \mu \alpha_4} (p_1 - p_5, -p_4, p_2 - p_3) & \Gamma_2^{A4} &= e \gamma_{\alpha_4} \\
\Gamma_1^{A5} &= \frac{g}{\sqrt{2}} \gamma_{\rho_2} P_L \Gamma_{\text{WWZ}}^{\rho_2 \mu \alpha_5} (p_1 - p_5, -p_4, p_2 - p_3) & \Gamma_2^{A5} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_5} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A6} &= \frac{g}{\sqrt{2}} \gamma^{\alpha_6} P_L & \Gamma_2^{A6} &= \frac{g^2}{2} \gamma_{\alpha_6} P_L (\not{p}_2 - \not{p}_4) \gamma^\mu P_L \tag{5} \\
\Gamma_1^{B1} &= e \gamma_{\beta_1} & \Gamma_2^{B1} &= \frac{g}{\sqrt{2}} \gamma_{\rho_3} P_L \Gamma_{\text{WWA}}^{\rho_3 \mu \beta_1} (p_2 - p_5, -p_4, p_1 - p_3) \\
\Gamma_1^{B2} &= e \gamma_{\beta_2} & \Gamma_2^{B2} &= \frac{eg}{\sqrt{2}} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma_{\beta_2} \\
\Gamma_1^{B3} &= \frac{g}{\cos \theta_W} \gamma_{\beta_3} (g_- P_L + g_+ P_R) & \Gamma_2^{B3} &= \frac{g}{\sqrt{2}} \gamma_{\rho_4} P_L \Gamma_{\text{WWZ}}^{\rho_4 \mu \beta_3} (p_2 - p_5, -p_4, p_1 - p_3) \\
\Gamma_1^{B4} &= \frac{g}{\cos \theta_W} \gamma_{\beta_4} (g_- P_L + g_+ P_R) & \Gamma_2^{B4} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma^{\beta_4} (g_- P_L + g_+ P_R) \\
\Gamma_1^{B5} &= \frac{g}{\cos \theta_W} \gamma_{\beta_5} (g_- P_L + g_+ P_R) & \Gamma_2^{B5} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^{\beta_5} P_L (\not{p}_2 - \not{p}_4) \gamma^\mu P_L
\end{aligned}$$

$$\Gamma_1^{B6} = \frac{g^2}{2} \gamma_{\beta_6} P_L (\not{p}_1 - \not{p}_4) \gamma^\mu P_L \quad \Gamma_2^{B6} = \frac{g}{\sqrt{2}} \gamma^{\beta_6} P_L \quad (6)$$

where we have defined

$$\begin{aligned} \Gamma_{\text{WWV}}^{\alpha,\beta,\mu}(p_{W^-}, p_{W^+}, p_A) &= g_{\text{WWV}} \left[(p_{W^-} - p_{W^+})^\mu + (p_{W^+} - p_A)^\alpha + (p_A - p_{W^-})^\beta \right] \\ g_{\text{WWA}} &= -e, \quad g_{\text{WWZ}} = g \cos \theta_W, \quad g_- = \sin^2 \theta_W, \quad g_+ = -\frac{1}{2} + \sin^2 \theta_W \end{aligned} \quad (7)$$

with the momenta entering the vertex. The denominators of the propagators are,

$$\begin{aligned} D_{A1} &= (-1)(-2p_2 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) & D_{A2} &= (-1)(-m_Z^2 - 2p_2 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) \\ D_{A3} &= (-1)(-m_Z^2 - 2p_2 \cdot p_3)(m_W^2 - 2p_1 \cdot p_4) & D_{A4} &= (-1)^2(-m_W^2 - 2p_1 \cdot p_5)(-2p_2 \cdot p_3) \\ D_{A5} &= (-1)^2(-m_W^2 - 2p_1 \cdot p_5)(-m_Z^2 - 2p_2 \cdot p_3) & D_{A6} &= (-1)(-m_W^2 - 2p_1 \cdot p_5)(m_W^2 - 2p_2 \cdot p_4) \\ D_{B1} &= (-1)^2(-2p_1 \cdot p_3)(-m_W^2 - 2p_2 \cdot p_5) & D_{B2} &= (-1)(-2p_1 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) \\ D_{B3} &= (-1)^2(-m_Z^2 - 2p_1 \cdot p_3)(-m_W^2 - 2p_2 \cdot p_5) & D_{B4} &= (-1)(-m_Z^2 - 2p_1 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) \\ D_{B5} &= (-1)(-m_Z^2 - 2p_1 \cdot p_3)(m_W^2 - 2p_2 \cdot p_4) & D_{B6} &= (-1)(m_W^2 - 2p_1 \cdot p_4)(-m_W^2 - 2p_2 \cdot p_5) \end{aligned} \quad (8)$$

We should note that in arriving at these expressions we took $m_e = 0$. It can be shown that the terms in the numerator of the propagators of the gauge bosons that are proportional to the momentum exactly vanish in this limit, and therefore we have put them to zero from the start. Also, we factor out the i^5 into the definition of iM . We keep the minus sign from the $-g^{\mu\nu}$ in the numerator of the vector boson propagators in the corresponding denominator.

C. The spin averaged $|M|^2$

The normal technique is to transform the spin averaged $|M|^2$ into traces. We get (we can take the denominators real, as there is no kinematical pole)

$$\begin{aligned} \overline{|M|^2} &= \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_5 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^A \right] \text{Tr} \left[\not{p}_3 \Gamma_{2i}^A \not{p}_2 \bar{\Gamma}_{2k}^A \right] \frac{1}{D_{Ai}} \frac{1}{D_{Ak}} + \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_3 \Gamma_{1i}^B \not{p}_1 \bar{\Gamma}_{1k}^B \right] \text{Tr} \left[\not{p}_5 \Gamma_{2i}^B \not{p}_2 \bar{\Gamma}_{2k}^B \right] \frac{1}{D_{Bi}} \frac{1}{D_{Bk}} \\ &\quad - \sum_{i,k=1}^6 \left(\text{Tr} \left[\not{p}_5 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^B \not{p}_3 \Gamma_{2i}^A \not{p}_2 \bar{\Gamma}_{2k}^B \right] + \text{Tr} \left[\not{p}_3 \Gamma_{1k}^B \not{p}_1 \bar{\Gamma}_{1i}^A \not{p}_5 \Gamma_{2k}^B \not{p}_2 \bar{\Gamma}_{2i}^A \right] \right) \frac{1}{D_{Ai}} \frac{1}{D_{Bk}} \end{aligned} \quad (9)$$

where, as usual, $\bar{\Gamma} = \gamma^0 (\Gamma)^\dagger \gamma^0$. Implicit is also the summation over the W polarizations,

$$\sum_{\lambda} \epsilon^\mu(k, \lambda) \epsilon^{\mu*}(k, \lambda) = -g^{\mu\nu} + \frac{k^\mu k^\nu}{m_W^2} \quad (10)$$

These traces are not so easy to evaluate because they are traces of a large number Dirac γ matrices. The best is to resort to automatic evaluation using some algebraic program. Nowadays, it seems to me that the more interesting approach is to use the package `FeynCalc` [1, 2] for `Mathematica`. In the appendix it is shown an example of a program that does that does this for this process. The code can be downloaded from my site [3].

III. $e^- e^+ \rightarrow W^- e^+ \nu_e$

A. Diagrams

This process corresponds to the Feynman diagrams of Fig. 2. Note, again, that there is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.




Figure 2: Feynman diagrams for $e^-e^+ \rightarrow e^+W^-\nu_e$. There is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.

B. Amplitudes

Again we write the amplitudes in the following form:

$$M_i = \sum_{i=1}^6 \bar{u}(p_5) \Gamma_1^{A_i} u(p_1) \bar{u}(p_3) \Gamma_2^{A_i} u(p_2) \frac{1}{D_{A_i}} - \sum_{i=1}^6 \bar{u}(p_3) \Gamma_1^{B_i} u(p_1) \bar{u}(p_5) \Gamma_2^{B_i} u(p_2) \frac{1}{D_{B_i}} \quad (11)$$

where we have put explicitly the minus sign between the two sets of diagrams, which we denote by A and B . We have

$$\begin{aligned}
\Gamma_1^{A1} &= \frac{eg}{\sqrt{2}} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma^{\alpha_1} & \Gamma_2^{A1} &= e \gamma_{\alpha_1} \\
\Gamma_1^{A2} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma^{\alpha_2} (g_- P_L + g_+ P_R) & \Gamma_2^{A2} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_2} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A3} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^{\alpha_3} P_L (\not{p}_1 - \not{p}_4) \gamma^\mu P_L & \Gamma_2^{A3} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_3} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A4} &= \frac{g}{\sqrt{2}} \gamma_{\rho_1} P_L \Gamma_{\text{WWA}}^{\rho_1 \mu \alpha_4} (p_1 - p_5, -p_4, p_2 - p_3) & \Gamma_2^{A4} &= e \gamma_{\alpha_4} \\
\Gamma_1^{A5} &= \frac{g}{\sqrt{2}} \gamma_{\rho_2} P_L \Gamma_{\text{WWZ}}^{\rho_2 \mu \alpha_5} (p_1 - p_5, -p_4, p_2 - p_3) & \Gamma_2^{A5} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_5} (g_- P_L + g_+ P_R) \\
\Gamma_1^{A6} &= \frac{g}{\sqrt{2}} \gamma^{\alpha_6} P_L & \Gamma_2^{A6} &= \frac{g^2}{2} \gamma_{\alpha_6} P_L (-\not{p}_3 - \not{p}_4) \gamma^\mu P_L
\end{aligned} \quad (12)$$

$$\begin{aligned}
\Gamma_1^{B1} &= e \gamma_{\beta_1} & \Gamma_2^{B1} &= \frac{g}{\sqrt{2}} \gamma_{\rho_3} P_L \Gamma_{\text{WWA}}^{\rho_3 \mu \beta_1} (-p_3 - p_5, -p_4, p_1 + p_2) \\
\Gamma_1^{B2} &= e \gamma^{\beta_2} & \Gamma_2^{B2} &= \frac{eg}{\sqrt{2}} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma_{\beta_2} \\
\Gamma_1^{B3} &= \frac{g}{\cos \theta_W} \gamma_{\beta_3} (g_- P_L + g_+ P_R) & \Gamma_2^{B3} &= \frac{g}{\sqrt{2}} \gamma_{\rho_4} \Gamma_{\text{WWZ}}^{\rho_4 \mu \beta_3} (-p_3 - p_5, -p_4, p_1 + p_2) \\
\Gamma_1^{B4} &= \frac{g}{\cos \theta_W} \gamma_{\beta_4} (g_- P_L + g_+ P_R) & \Gamma_2^{B4} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^\mu P_L (\not{p}_4 + \not{p}_5) \gamma^{\beta_4} (g_- P_L + g_+ P_R) \\
\Gamma_1^{B5} &= \frac{g}{\cos \theta_W} \gamma_{\beta_5} (g_- P_L + g_+ P_R) & \Gamma_2^{B5} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^{\beta_5} P_L (-\not{p}_3 - \not{p}_4) \gamma^\mu P_L \\
\Gamma_1^{B6} &= \frac{g^2}{2} \gamma_{\beta_6} P_L (\not{p}_1 - \not{p}_4) \gamma^\mu P_L & \Gamma_2^{B6} &= \frac{g}{\sqrt{2}} \gamma^{\beta_6} P_L
\end{aligned} \tag{13}$$

The denominators of the propagators are,

$$\begin{aligned}
D_{A1} &= (-1)(-2p_2 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) & D_{A2} &= (-1)(-m_Z^2 - 2p_2 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) \\
D_{A3} &= (-1)(-m_Z^2 - 2p_2 \cdot p_3)(m_W^2 - 2p_1 \cdot p_4) & D_{A4} &= (-1)^2(-m_W^2 - 2p_1 \cdot p_5)(-2p_2 \cdot p_3) \\
D_{A5} &= (-1)^2(-m_W^2 - 2p_1 \cdot p_5)(-m_Z^2 - 2p_2 \cdot p_3) & D_{A6} &= (-1)(-m_W^2 - 2p_1 \cdot p_5)(m_W^2 + 2p_3 \cdot p_4) \\
D_{B1} &= (-1)^2(2p_1 \cdot p_2)(-m_W^2 + 2p_3 \cdot p_5 + i m_W \Gamma_W) & D_{B2} &= (-1)(2p_1 \cdot p_2)(m_W^2 + 2p_4 \cdot p_5) \\
D_{B3} &= (-1)^2(2p_1 \cdot p_2 - m_Z^2 + i m_Z \Gamma_Z)(2p_3 \cdot p_5 - m_W^2 + i m_W \Gamma_W) & D_{B4} &= (-1)(-m_Z^2 + 2p_1 \cdot p_2 + i m_Z \Gamma_Z)(m_W^2 + 2p_4 \cdot p_5) \\
D_{B5} &= (-1)(-m_Z^2 + 2p_1 \cdot p_2 + i m_Z \Gamma_Z)(m_W^2 + 2p_3 \cdot p_4) & D_{B6} &= (-1)(m_W^2 - 2p_1 \cdot p_4)(2p_3 \cdot p_5 - m_W^2 + i m_W \Gamma_W)
\end{aligned} \tag{14}$$

C. The spin averaged $|M|^2$

The normal technique is to transform the spin averaged $|M|^2$ into traces. The diagrams of type A have no kinematical pole and the corresponding propagators can be taken real. However, those of type B have now poles and they should be taken as complex with the inclusion of the widths. We get,

$$\begin{aligned}
\overline{|M|^2} &= \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_5 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^A \right] \text{Tr} \left[\not{p}_2 \Gamma_{2i}^A \not{p}_3 \bar{\Gamma}_{2k}^A \right] \frac{1}{D_{Ai}} \frac{1}{D_{Ak}} + \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_2 \Gamma_{1i}^B \not{p}_1 \bar{\Gamma}_{1k}^B \right] \text{Tr} \left[\not{p}_5 \Gamma_{2i}^B \not{p}_3 \bar{\Gamma}_{2k}^B \right] \frac{1}{D_{Bi}} \frac{1}{D_{Bk}^*} \\
&\quad - \sum_{i,k=1}^6 \left(\text{Tr} \left[\not{p}_5 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^B \not{p}_2 \Gamma_{2i}^A \not{p}_3 \bar{\Gamma}_{2k}^B \right] \frac{1}{D_{Bk}^*} + \text{Tr} \left[\not{p}_2 \Gamma_{1k}^B \not{p}_1 \bar{\Gamma}_{1i}^A \not{p}_5 \Gamma_{2k}^B \not{p}_3 \bar{\Gamma}_{2i}^A \right] \frac{1}{D_{Bk}} \right) \frac{1}{D_{Ai}}
\end{aligned} \tag{15}$$

where, as usual, $\bar{\Gamma} = \gamma^0 (\Gamma)^\dagger \gamma^0$. Implicit is also the summation over the W polarizations,

$$\sum_{\lambda} \epsilon^\mu(k, \lambda) \epsilon^{\mu*}(k, \lambda) = -g^{\mu\nu} + \frac{k^\mu k^\nu}{m_W^2} \tag{16}$$

These traces are not so easy to evaluate because they are traces of a large number Dirac γ matrices. The code for `FeunCalc` can be downloaded from my site [3].

IV. $e^- e^+ \rightarrow W^+ e^- \bar{\nu}_e$

A. Diagrams

This process corresponds to the Feynman diagrams of Fig. 3. Note, again, that there is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.




Figure 3: Feynman diagrams for $e^-e^+ \rightarrow e^-W^+\bar{\nu}_e$. There is relative minus sign between the first six diagrams (A_1, \dots, A_6) and the last six diagrams (B_1, \dots, B_6), corresponding to the interchange of a fermionic line.

B. Amplitudes

Again we write the amplitudes in the following form:

$$M_i = \sum_{i=1}^6 \bar{u}(p_5) \Gamma_1^{Ai} u(p_1) \bar{u}(p_3) \Gamma_2^{Ai} u(p_2) \frac{1}{D_{Ai}} - \sum_{i=1}^6 \bar{u}(p_3) \Gamma_1^{Bi} u(p_1) \bar{u}(p_5) \Gamma_2^{Bi} u(p_2) \frac{1}{D_{Bi}} \quad (17)$$

where we have put explicitly the minus sign between the two sets of diagrams, which we denote by A and B . We have

$$\begin{aligned}
\Gamma_1^{A1} &= e \gamma_{\alpha_1} & \Gamma_2^{A1} &= \frac{eg}{\sqrt{2}} \gamma^{\alpha_1} (-\not{p}_4 - \not{p}_5) \gamma^\mu P_L \\
\Gamma_1^{A2} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_2} (g_- P_L + g_+ P_R) & \Gamma_2^{A2} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^{\alpha_2} (g_- P_L + g_+ P_R) (-\not{p}_4 - \not{p}_5) \gamma^\mu P_L \\
\Gamma_1^{A3} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_3} (g_- P_L + g_+ P_R) & \Gamma_2^{A3} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^\mu P_L (-\not{p}_2 + \not{p}_4) \gamma^{\alpha_3} P_L \\
\Gamma_1^{A4} &= e \gamma_{\alpha_4} & \Gamma_2^{A4} &= \frac{g}{\sqrt{2}} \gamma_{\rho_1} P_L \Gamma_{\text{WWA}}^{\mu \rho_1 \alpha_4} (-p_4, p_2 - p_5, p_1 - p_3) \\
\Gamma_1^{A5} &= \frac{g}{\cos \theta_W} \gamma_{\alpha_5} (g_- P_L + g_+ P_R) & \Gamma_2^{A5} &= \frac{g}{\sqrt{2}} \gamma_{\rho_2} P_L \Gamma_{\text{WWZ}}^{\mu \rho_2 \alpha_5} (-p_4, p_2 - p_5, p_1 - p_3) \\
\Gamma_1^{A6} &= \frac{g^2}{2} \gamma^\mu P_L (\not{p}_3 + \not{p}_4) \gamma_{\alpha_6} P_L & \Gamma_2^{A6} &= \frac{g}{\sqrt{2}} \gamma^{\alpha_6} P_L \\
\Gamma_1^{B1} &= e \gamma_{\beta_1} & \Gamma_2^{B1} &= \frac{g}{\sqrt{2}} \gamma_{\rho_3} \Gamma_{\text{WWA}}^{\mu \rho_3 \beta_1} (-p_4, -p_3 - p_5, p_1 + p_2)
\end{aligned} \quad (18)$$

$$\begin{aligned}
\Gamma_1^{B2} &= e \gamma^{\beta_2} & \Gamma_2^{B2} &= \frac{eg}{\sqrt{2}} \gamma_{\beta_2} (-\not{p}_4 - \not{p}_5) \gamma^\mu P_L \\
\Gamma_1^{B3} &= \frac{g}{\cos \theta_W} \gamma_{\beta_3} (g_- P_L + g_+ P_R) & \Gamma_2^{B3} &= \frac{g}{\sqrt{2}} \gamma_{\rho_4} \Gamma_{\text{WWZ}}^{\mu \rho_4 \beta_3} (-p_4, -p_3 - p_5, p_1 + p_2) \\
\Gamma_1^{B4} &= \frac{g}{\cos \theta_W} \gamma_{\beta_4} (g_- P_L + g_+ P_R) & \Gamma_2^{B4} &= \frac{g^2}{\sqrt{2} \cos \theta_W} \gamma^{\beta_4} (g_- P_L + g_+ P_R) (-\not{p}_4 - \not{p}_5) \gamma^\mu P_L \\
\Gamma_1^{B5} &= \frac{g}{\cos \theta_W} \gamma_{\beta_5} (g_- P_L + g_+ P_R) & \Gamma_2^{B5} &= \frac{g^2}{2\sqrt{2} \cos \theta_W} \gamma^\mu P_L (\not{p}_3 + \not{p}_4) \gamma^{\beta_5} P_L \\
\Gamma_1^{B6} &= \frac{g^2}{2} \gamma^\mu P_L (-\not{p}_2 + \not{p}_4) \gamma_{\beta_6} P_L & \Gamma_2^{B6} &= \frac{g}{\sqrt{2}} \gamma^{\beta_6} P_L
\end{aligned} \tag{19}$$

The denominators of the propagators are,

$$\begin{aligned}
D_{A1} &= (-1)(-2p_1 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) & D_{A2} &= (-1)(-m_Z^2 - 2p_1 \cdot p_3)(m_W^2 + 2p_4 \cdot p_5) \\
D_{A3} &= (-1)(-m_Z^2 - 2p_1 \cdot p_3)(m_W^2 - 2p_2 \cdot p_4) & D_{A4} &= (-1)^2(-m_W^2 - 2p_2 \cdot p_5)(-2p_1 \cdot p_3) \\
D_{A5} &= (-1)^2(-m_W^2 - 2p_2 \cdot p_5)(-m_Z^2 - 2p_1 \cdot p_3) & D_{A6} &= (-1)(-m_W^2 - 2p_2 \cdot p_5)(m_W^2 + 2p_3 \cdot p_4) \\
D_{B1} &= (-1)^2(2p_1 \cdot p_2)(-m_W^2 + 2p_3 \cdot p_5 + im_W \Gamma_W) & D_{B2} &= (-1)(2p_1 \cdot p_2)(m_W^2 + 2p_4 \cdot p_5) \\
D_{B3} &= (-1)^2(2p_1 \cdot p_2 - m_Z^2 + im_Z \Gamma_Z)(2p_3 \cdot p_5 - m_W^2 + im_W \Gamma_W) & D_{B4} &= (-1)(2p_1 \cdot p_2 - m_Z^2 + im_Z \Gamma_Z)(m_W^2 + 2p_4 \cdot p_5) \\
D_{B5} &= (-1)(-m_Z^2 + 2p_1 \cdot p_2 + im_Z \Gamma_Z)(m_W^2 + 2p_3 \cdot p_4) & D_{B6} &= (-1)(m_W^2 - 2p_2 \cdot p_4)(2p_3 \cdot p_5 - m_W^2 + im_W \Gamma_W)
\end{aligned} \tag{20}$$

C. The spin averaged $|M|^2$

The normal technique is to transform the spin averaged $|M|^2$ into traces. The diagrams of type *A* have no kinematical pole and the corresponding propagators can be taken real. However, those of type *B* have now poles and they should be taken as complex with the inclusion of the widths. We get,

$$\begin{aligned}
\overline{|M|^2} &= \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_3 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^A \right] \text{Tr} \left[\not{p}_2 \Gamma_{2i}^A \not{p}_5 \bar{\Gamma}_{2k}^A \right] \frac{1}{D_{Ai}} \frac{1}{D_{Ak}} + \sum_{i,k=1}^6 \text{Tr} \left[\not{p}_2 \Gamma_{1i}^B \not{p}_1 \bar{\Gamma}_{1k}^B \right] \text{Tr} \left[\not{p}_3 \Gamma_{2i}^B \not{p}_5 \bar{\Gamma}_{2k}^B \right] \frac{1}{D_{Bi}} \frac{1}{D_{Bk}^*} \\
&\quad - \sum_{i,k=1}^6 \left(\text{Tr} \left[\not{p}_3 \Gamma_{1i}^A \not{p}_1 \bar{\Gamma}_{1k}^B \not{p}_2 \Gamma_{2i}^A \not{p}_5 \bar{\Gamma}_{2k}^B \right] \frac{1}{D_{Bk}^*} + \text{Tr} \left[\not{p}_2 \Gamma_{1k}^B \not{p}_1 \bar{\Gamma}_{1i}^A \not{p}_3 \Gamma_{2k}^B \not{p}_5 \bar{\Gamma}_{2i}^A \right] \frac{1}{D_{Bk}} \right) \frac{1}{D_{Ai}}
\end{aligned} \tag{21}$$

where, as usual, $\bar{\Gamma} = \gamma^0 (\Gamma)^\dagger \gamma^0$. Implicit is also the summation over the *W* polarizations,

$$\sum_{\lambda} \epsilon^\mu(k, \lambda) \epsilon^{\mu*}(k, \lambda) = -g^{\mu\nu} + \frac{k^\mu k^\nu}{m_W^2} \tag{22}$$

These traces are not so easy to evaluate because they are traces of a large number Dirac γ matrices. The code for `FeunCalc` can be downloaded from my site [3].

V. THE CROSS SECTION

We choose the kinematics in the CM frame shown in Fig 4. The cross section is then given by,

$$d\sigma = \frac{1}{(2\pi)^5} \frac{1}{2s} \overline{|M|^2} \delta^4(p_1 + p_2 - p_3 - p_4 - p_5) \frac{d^3 p_3}{2p_3^0} \frac{d^3 p_4}{2p_4^0} \frac{d^3 p_5}{2p_5^0} \tag{23}$$

This can be written as,

$$\frac{d\sigma}{dE_e d\cos\theta} = \frac{E_3}{64\pi^4 s} \int_0^{2\pi} \frac{d\varphi}{2\pi} \int \frac{d^3 q_1}{2q_1^0} \frac{d^3 q_2}{2q_2^0} \delta^4(p_1 + p_2 - p_3 - p_4 - p_5) \overline{|M|^2}. \tag{24}$$




Figure 4: Kinematics for $e^-e^- \rightarrow W^-e^-\nu_e$ and related processes.

To proceed we must do the integrations over the two remaining momenta. For this we evaluate the following quantity,

$$X = \int \frac{d^3q_1}{2q_1^0} \frac{d^3q_2}{2q_2^0} \delta^4(\Delta - p_4 - p_5) \overline{|M|^2}. \quad (25)$$

where $\Delta = p_1 + p_2 - p_3$. These quantities are Lorentz invariant and can be evaluated in any reference frame. However the calculation it is particularly simple in the CM frame of the two remaining particles. In that frame

$$\Delta = (m_{45}, 0, 0, 0), \quad (26)$$

where $m_{45}^2 = (p_4 + p_5)^2 = (p_1 + p_2 - p_3)^2$ is the invariant mass of the pair. We then get (p_4^*, p_5^* are the vectors in the CM frame of the pair)

$$\begin{aligned} X &= \frac{1}{4} \int \frac{d^3p_4^*}{p_4^{*0} p_5^{*0}} \delta(m_{45} - p_4^{*0} - p_5^{*0}) \overline{|M|^2} \\ &= \frac{1}{4} \int \frac{|\vec{p}_4^*| dp_4^{*0} d\Omega_4^*}{p_5^{*0}} \delta\left(m_{45} - p_4^{*0} - \sqrt{(p_4^{*0})^2 - m_W^2}\right) \overline{|M|^2} \\ &= \frac{1}{8} \left(1 - \frac{m_W^2}{m_{45}^2}\right) \int d\Omega_4^* \overline{|M|^2} \end{aligned} \quad (27)$$

and the final form for the cross section is,

$$\sigma = \frac{1}{64\pi^4 s} \int_0^{E_3^{\max}} dE_3 E_3 \int_0^\pi d\theta \sin\theta \int_0^\pi d\theta^* \sin\theta^* \int_0^{2\pi} d\varphi^* \frac{1}{8} \left(1 - \frac{m_W^2}{m_{45}^2}\right) \overline{|M|^2} \quad (28)$$

and we have already used the fact that the final result does not depend on the azimuthal angle φ , and have done that integration. The maximum energy of the electron is given by, $E_3^{\max} = (s - m_W^2)/(2s)$. It is crucial to understand the meaning of the angles in the previous expression. The angle θ is the scattering angle of the electron with momentum p_3 in the CM of the e^-e^+ system (the lab frame), while the angles (θ^*, φ^*) are the angles of the W boson (with momentum p_4) in the CM frame of the pair p_4, p_5 . It is in these variables that the integrations are simple to set, but of course, there remains the difficulty of relating the 4-momenta in that frame to the lab frame. We will discuss in generality this question in the following section.

VI. KINEMATICS: HOW TO RELATE DIFFERENT REFERENCE FRAMES

As we have seen in the previous section we have to relate the coordinates of 4-vectors in different reference frames. The best way is to do this in steps, using in sequence elementary rotations and boosts. We always perform boosts along the z axis, so the relevant Lorentz transformation is

$$x = \mathbf{Boost}_z(\beta) \cdot x' \quad (29)$$

where $x(x')$ are the coordinates in the reference frame $S(S')$, respectively, and the reference frame S' moves with relative speed β along the z axis with respect to S . The explicit form of the matrix $\mathbf{Boost}_z(\beta)$ is

$$\mathbf{Boost}_z(\beta) = \begin{bmatrix} \gamma & 0 & 0 & \gamma\beta \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \gamma\beta & 0 & 0 & \gamma \end{bmatrix} \quad (30)$$

For rotations, we can always use a sequence of rotations along the coordinate axis. So the relevant rotations, always written with the convention

$$x = \mathbf{Rot}_x(\theta) \cdot x' \quad ; \quad x = \mathbf{Rot}_y(\theta) \cdot x' \quad ; \quad x = \mathbf{Rot}_z(\theta) \cdot x' \quad (31)$$

are

$$\mathbf{Rot}_x(\theta) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \cos \theta & \sin \theta \\ 0 & 0 & -\sin \theta & \cos \theta \end{bmatrix} \quad \mathbf{Rot}_y(\theta) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \cos \theta & 0 & \sin \theta \\ 0 & 0 & 1 & 0 \\ 0 & -\sin \theta & 0 & \cos \theta \end{bmatrix} \quad (32)$$

and

$$\mathbf{Rot}_z(\theta) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \cos \theta & \sin \theta & 0 \\ 0 & -\sin \theta & \cos \theta & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (33)$$

Now, if we look at the kinematics shown in Fig. 4, we can easily realize that the correct transformation laws for p_4^* and p_5^* in the CM frame of the Z boson pair are,

$$p_{4,5} = \mathbf{Rot}_y(\theta + \pi) \cdot \mathbf{Boost}_z(\beta) \cdot p_{4,5}^* \quad (34)$$

where

$$\begin{aligned} \beta &= \sqrt{1 - \frac{1}{\gamma^2}} \\ \gamma &= \frac{E_{45}}{m_{45}} \\ m_{45} &= \sqrt{(p_4 + p_5)^2} = \sqrt{s - 2\sqrt{s}E_3} \\ E_{45} &= \frac{s + m_{45}^2}{2\sqrt{s}} \end{aligned} \quad (35)$$

Taking in account that

$$\begin{cases} p_4^{*0} = \frac{m_{45}^2 + m_W^2}{2m_{45}} \\ p_4^{*1} = p_{45}^{\text{CM}} \sin \theta^* \cos \varphi^* \\ p_4^{*2} = p_{45}^{\text{CM}} \sin \theta^* \sin \varphi^* \\ p_4^{*3} = p_{45}^{\text{CM}} \cos \theta^* \end{cases} \quad \begin{cases} p_5^{*0} = \frac{m_{45}^2 - m_W^2}{2m_{45}} \\ p_5^{*1} = -p_{45}^{\text{CM}} \sin \theta^* \cos \varphi^* \\ p_5^{*2} = -p_{45}^{\text{CM}} \sin \theta^* \sin \varphi^* \\ p_5^{*3} = -p_{45}^{\text{CM}} \cos \theta^* \end{cases} \quad (36)$$

where

$$p_{45}^{\text{CM}} = \frac{m_{45}^2 - m_W^2}{2m_{45}} \quad (37)$$

we can now use Eq. (34) to obtain p_4 and p_5 in the lab frame, as a function of $(E_3, \theta, \theta^*, \varphi^*)$.

VII. NUMERICAL INTEGRATION

With the kinematics completely implemented, we have to perform the integrations and obtain the cross sections. In few problems can the integrals that appear in the evaluation of cross sections be done analytically. Most of the time we have to revert to numerical methods to evaluate the integrals. There are many ways to do evaluate the integrals numerically. A good library is `CUBA package`[4]. It can be linked either with `C/C++` or with `Fortran` programs. It has several methods with the same calling structure, so that one can substitute one method for another and check for accuracy. The manual comes with the package. I have done a gaussian integration subroutine in `Fortran` called `gauss.f` that uses the same calling conventions, and therefore can also be tested in the same way. The codes for these programs can be found in a Web page that I have created [3].

A. $e^-e^- \rightarrow W^-e^-\nu_e$

We have calculated both the unpolarized cross section and the cases when the beams have L or R polarization. For this we just have to substitute in Eq. (21)

$$p_1 \rightarrow (aP_L + bP_R), \quad p_2 \rightarrow (cP_L + dP_R) \quad (38)$$

The results are shown in Fig. 5 where we also show the results obtained with CalcHep[5] for the unpolarized cross section. We note that due to the collinear divergence we have to apply a cut on the θ angle. For all cases we took $|\cos\theta| < 0.985$.




Figure 5: Cross section for $e^-e^- \rightarrow W^-e^-\nu_e$. The blue crosses are for the results of CalcHep for the unpolarized cross section.

B. $e^-e^+ \rightarrow W^-e^+\nu_e$

The results are shown in Fig. 6 where we also show the results obtained with CalcHep[5] for the unpolarized cross section. On the right panel of the the Figure we show in more detail the behavior for low \sqrt{s} where the widths in the W and Z propagator are necessary for the diagrams of type B in Fig. 2. We see that the agreement is excellent.

C. $e^-e^+ \rightarrow W^+e^-\bar{\nu}_e$

The results are shown in Fig. 7 where we also show the results obtained with CalcHep[5] for the unpolarized cross section. On the right panel of the the Figure we show in more detail the behavior for low \sqrt{s} where the widths in the W and Z propagator are necessary for the diagrams of type B in Fig. 2. Again we see that the agreement is excellent. Also it is clear the this process should be related to the previous one $e^-e^+ \rightarrow W^-e^+\nu_e$. The following relation hold:

$$\begin{aligned} \sigma_{LL}(e^-e^+ \rightarrow W^+e^-\bar{\nu}_e) &= \sigma_{RR}(e^-e^+ \rightarrow W^-e^+\nu_e) \\ \sigma_{RR}(e^-e^+ \rightarrow W^+e^-\bar{\nu}_e) &= \sigma_{LR}(e^-e^+ \rightarrow W^-e^+\nu_e) \\ \sigma_{LR}(e^-e^+ \rightarrow W^+e^-\bar{\nu}_e) &= \sigma_{LR}(e^-e^+ \rightarrow W^-e^+\nu_e) \\ \sigma_{RL}(e^-e^+ \rightarrow W^+e^-\bar{\nu}_e) &= \sigma_{RL}(e^-e^+ \rightarrow W^-e^+\nu_e) \end{aligned} \quad (39)$$

This can also be verified in Fig 7




Figure 6: Cross section for $e^-e^+ \rightarrow W^-e^+\nu_e$. The blue crosses or dots are for the results of CalcHep for the unpolarized cross section.




Figure 7: Cross section for $e^-e^+ \rightarrow W^+e^-\bar{\nu}_e$. The blue crosses or dots are for the results of CalcHep for the unpolarized cross section. The other crosses are for comparison with the process $e^-e^+ \rightarrow W^-e^+\nu_e$.

VIII. CONCLUSIONS

In this note we review the techniques used in Quantum Field Theory in a set of non trivial examples, the scatterings $e^-e^- \rightarrow W^-e^-\nu_e$, $e^-e^+ \rightarrow W^-e^+\nu_e$ and $e^-e^+ \rightarrow W^+e^-\bar{\nu}_e$. The emphasis is in the use of the package `FeynCalc` for `Mathematica`. Comparison is made with CalcHep[5] for the unpolarized cross section.

Appendix A: INPUT FOR FEYNCALC

We give here the code for the process $e^-e^- \rightarrow W^-e^-\nu_e$. It can be downloaded from my web page [3], where you also find the code for the other related processes.

```

(***** Program eeWene.m *****)
(* Program to calculate the amplitudes squared for the process

   e^- e^- -> W^- e^- \nu_e

in the SM
*)

Remove["Global' *"]
Batch=True
If [Batch, <<FC]
TimeNow=TimeUsed[];

(* Definitons *)

SumPolV=Function[{k,m,mu,nu}, -MT[mu,nu]+FV[k,mu]*FV[k,nu]/m^2];
PL:=DiracGamma[7];
PR:=DiracGamma[6];

p5=p1+p2-p3-p4;

kin={SP[p1,p1]->0,SP[p2,p2]->0,SP[p3,p3]->0,SP[p4,p4]->mW^2,
SP[p3,p4]->(-mW^2 -2 SP[p1,p2]+ 2 SP[p1,p3] + 2 SP[p1,p4]+ 2 SP[p2,p3]+2 SP[p2,p4])/2};

vecmom={SP[p1,p2]->p1p2,SP[p1,p3]->p1p3,SP[p1,p4]->p1p4,
SP[p2,p3]->p2p3,SP[p2,p4]->p2p4,SP[p3,p4]->p3p4,LC[] [p1, p2, p3, p4]->Epsp1p2p3p4};

kinFCI={Pair[Momentum[p1], Momentum[p1]] -> 0,
Pair[Momentum[p2], Momentum[p2]] -> 0,
Pair[Momentum[p3], Momentum[p3]] -> 0,
Pair[Momentum[p4], Momentum[p4]] -> mW^2,
Pair[Momentum[p3], Momentum[p4]] ->
(-mW^2 - 2*Pair[Momentum[p1], Momentum[p2]] +
2*Pair[Momentum[p1], Momentum[p3]] +
2*Pair[Momentum[p1], Momentum[p4]] +
2*Pair[Momentum[p2], Momentum[p3]] +
2*Pair[Momentum[p2], Momentum[p4]])/2,
gWV[A]->gWWA,gWV[Z]->gWWZ,gW->g2/Sqrt[2]};

Pol=Function[{a,b},a PL + b PR];

VMMWPA=Function[{a1,be,ga,p1,p2,p3,V},
gWV[V] (MT[a1,be] ((FourVector[p1,ga]) - (FourVector[p2,ga])) +
MT[be,ga] ((FourVector[p2,a1]) - (FourVector[p3,a1])) +
MT[ga,a1] ((FourVector[p3,be]) - (FourVector[p1,be]))));

stmpAll=OpenWrite["Diags.f",FormatType->FortranForm,PageWidth->68];
stmpImag=OpenWrite["Imag.f",FormatType->FortranForm,PageWidth->68];

(* Propagators *)

SignVB=-1
SymmetryEven=+1
SymmetryOdd=-1

DenA1A=(-2 SP[p2,p3]) SignVB
DenA1E1=(mW^2+2 SP[p4,p5])
DenA1 = Calc[DenA1A DenA1E1 SymmetryEven] /.kinFCI//Simplify

```

```

(***** Program eeWene.m (Continuation) *****)
DenA2Z=(-mZ^2-2 SP[p2,p3]) SignVB
DenA2E1=(mW^2+2 SP[p4,p5])
DenA2 = Calc[DenA2Z DenA2E1 SymmetryEven] /.kinFCI//Simplify

DenA3Nu = (mW^2 -2 SP[p1,p4])
DenA3Z = (-mZ^2 -2 SP[p2,p3]) SignVB
DenA3 = Calc[DenA3Nu DenA3Z SymmetryEven] /.kinFCI//Simplify

DenA4W = (-mW^2 -2 SP[p1,p5]) SignVB
DenA4A = (-2 SP[p2,p3]) SignVB
DenA4 = Calc[DenA4W DenA4A SymmetryEven] /.kinFCI//Simplify

DenA5W = (-mW^2 -2 SP[p1,p5]) SignVB
DenA5Z = (-mZ^2 -2 SP[p2,p3]) SignVB
DenA5 = Calc[DenA5W DenA5Z SymmetryEven] /.kinFCI//Simplify

DenA6W = (-mW^2 -2 SP[p1,p5]) SignVB
DenA6Nu = (mW^2 -2 SP[p2,p4])
DenA6 = Calc[DenA6W DenA6Nu SymmetryEven] /.kinFCI//Simplify

DenB1A = (-2 SP[p1,p3]) SignVB
DenB1W = (-mW^2 -2 SP[p2,p5]) SignVB
DenB1 = Calc[DenB1A DenB1W SymmetryOdd] /.kinFCI//Simplify

DenB2A = (-2 SP[p1,p3]) SignVB
DenB2E1 = (mW^2 + 2 SP[p4,p5])
DenB2 = Calc[DenB2A DenB2E1 SymmetryOdd] /.kinFCI//Simplify

DenB3Z = (-mZ^2 - 2 SP[p1,p3]) SignVB
DenB3W = (-mW^2 -2 SP[p2,p5]) SignVB
DenB3 = Calc[DenB3Z DenB3W SymmetryOdd] /.kinFCI//Simplify

DenB4Z = (-mZ^2 - 2 SP[p1,p3]) SignVB
DenB4E1 = (mW^2 + 2 SP[p4,p5])
DenB4 = Calc[DenB4Z DenB4E1 SymmetryOdd] /.kinFCI//Simplify

DenB5Z = (-mZ^2 - 2 SP[p1,p3]) SignVB
DenB5Nu = (mW^2 - 2 SP[p2,p4])
DenB5 = Calc[DenB5Z DenB5Nu SymmetryOdd] /.kinFCI//Simplify

DenB6Nu = (mW^2 -2 SP[p1,p4])
DenB6W = (-mW^2 -2 SP[p2,p5]) SignVB
DenB6 = Calc[DenB6Nu DenB6W SymmetryOdd] /.kinFCI//Simplify

DenA={DenA1,DenA2,DenA3,DenA4,DenA5,DenA6};
DenB={DenB1,DenB2,DenB3,DenB4,DenB5,DenB6};
DenBc=DenB

```

```

(***** Program eeWene.m (Continuation) *****)
(* Definiton of fermion lines *)
(* Gammas *)

Gamma1A1 = GA[mu] . (gW PL) . (GS[p4]+GS[p5]) . (ee GA[id1]) ;
Gamma2A1 = (ee GA[id1]) ;
Gamma1A1c = (ee GA[id2]) . (GS[p4]+GS[p5]) . GA[mup] . (gW PL);
Gamma2A1c = (ee GA[id2]) ;

Gamma1A2 = GA[mu] .(gW PL) . (GS[p4]+GS[p5]) . GA[id3] . (gZL PL + gZR PR);
Gamma2A2 = GA[id3] . ( gZL PL + gZR PR);
Gamma1A2c = GA[id4] . (gZL PL + gZR PR) . (GS[p4]+GS[p5]) . GA[mup] .(gW PL);
Gamma2A2c = GA[id4] . ( gZL PL + gZR PR);

Gamma1A3 =GA[id5] .(gZnuL PL + gZnuR PR) . (GS[p1]-GS[p4]). GA[mu] . (gW PL);
Gamma2A3 = GA[id5] . ( gZL PL + gZR PR);
Gamma1A3c= GA[mup] . (gW PL).(GS[p1]-GS[p4]). GA[id6] .(gZnuL PL + gZnuR PR);
Gamma2A3c =GA[id6] . ( gZL PL + gZR PR);

Gamma1A4 = GA[a1] . (gW PL) VWMWPA[a1,mu,be,p1-p5,-p4,p2-p3,A];
Gamma2A4 = (ee GA[be]);
Gamma1A4c = GA[alp] . (gW PL) VWMWPA[alp,mup,bep,p1-p5,-p4,p2-p3,A];
Gamma2A4c = (ee GA[bep]);

Gamma1A5 = GA[a1] . (gW PL) VWMWPA[a1,mu,be,p1-p5,-p4,p2-p3,Z];
Gamma2A5 = GA[be] . ( gZL PL + gZR PR);
Gamma1A5c= GA[alp] . (gW PL) VWMWPA[alp,mup,bep,p1-p5,-p4,p2-p3,Z];
Gamma2A5c= GA[bep] . ( gZL PL + gZR PR);

Gamma1A6 = GA[id7] . (gW PL);
Gamma2A6 = GA[id7] . (gW PL) . (GS[p2]-GS[p4]) . GA[mu] . (gW PL);
Gamma1A6c = GA[id8] . (gW PL);
Gamma2A6c = GA[mup] . (gW PL) . (GS[p2]-GS[p4]) . GA[id8] . (gW PL) ;

Gamma1B1 = (ee GA[id9]);
Gamma2B1 = GA[id10] . (gW PL) VWMWPA[id10,mu,id9,p2-p5,-p4,p1-p3,A];
Gamma1B1c = (ee GA[id11]);
Gamma2B1c = GA[id12] . (gW PL) VWMWPA[id12,mup,id11,p2-p5,-p4,p1-p3,A];

Gamma1B2 = (ee GA[id13]);
Gamma2B2 = GA[mu] . (gW PL) . (GS[p4]+GS[p5]) . (ee GA[id13]);
Gamma1B2c = (ee GA[id14]);
Gamma2B2c = (ee GA[id14]) . (GS[p4]+GS[p5]) . GA[mup] . (gW PL) ;

Gamma1B3 = GA[id15] . (gZL PL + gZR PR);
Gamma2B3 = GA[id16] . (gW PL) VWMWPA[id16,mu,id15,p2-p5,-p4,p1-p3,Z];
Gamma1B3c = GA[id17] . (gZL PL + gZR PR);
Gamma2B3c = GA[id18] . (gW PL) VWMWPA[id18,mup,id17,p2-p5,-p4,p1-p3,Z];

Gamma1B4 = GA[id19] . (gZL PL + gZR PR);
Gamma2B4 = GA[mu] . (gW PL) . (GS[p4]+GS[p5]) . GA[id19] . (gZL PL + gZR PR);
Gamma1B4c = GA[id20] . (gZL PL + gZR PR);
Gamma2B4c = GA[id20] . (gZL PL + gZR PR) . (GS[p4]+GS[p5]) . GA[mup] . (gW PL);

Gamma1B5 = GA[id21] . (gZL PL + gZR PR);
Gamma2B5 = GA[id21] . (gZnuL PL + gZnuR PR).(GS[p2]-GS[p4]) . GA[mu] . (gW PL);
Gamma1B5c = GA[id22] . (gZL PL + gZR PR);
Gamma2B5c = GA[mup] . (gW PL).(GS[p2]-GS[p4]) . GA[id22].(gZnuL PL + gZnuR PR);

```

```

(***** Program eeWene.m (Continuation) *****)
Gamma1B6 = GA[id23] . (gW PL) . (GS[p1]-GS[p4]) . GA[mu] . (gW PL);
Gamma2B6 = GA[id23] . (gW PL);
Gamma1B6c = GA[mup] . (gW PL) . (GS[p1]-GS[p4]) . GA[id24] . (gW PL);
Gamma2B6c = GA[id24] . (gW PL);

Gamma1A={Gamma1A1,Gamma1A2,Gamma1A3,Gamma1A4,Gamma1A5,Gamma1A6};
Gamma2A={Gamma2A1,Gamma2A2,Gamma2A3,Gamma2A4,Gamma2A5,Gamma2A6};
Gamma1Ac={Gamma1A1c,Gamma1A2c,Gamma1A3c,Gamma1A4c,Gamma1A5c,Gamma1A6c};
Gamma2Ac={Gamma2A1c,Gamma2A2c,Gamma2A3c,Gamma2A4c,Gamma2A5c,Gamma2A6c};

Gamma1B={Gamma1B1,Gamma1B2,Gamma1B3,Gamma1B4,Gamma1B5,Gamma1B6};
Gamma2B={Gamma2B1,Gamma2B2,Gamma2B3,Gamma2B4,Gamma2B5,Gamma2B6};
Gamma1Bc={Gamma1B1c,Gamma1B2c,Gamma1B3c,Gamma1B4c,Gamma1B5c,Gamma1B6c};
Gamma2Bc={Gamma2B1c,Gamma2B2c,Gamma2B3c,Gamma2B4c,Gamma2B5c,Gamma2B6c};

If[False,

(* Evaluate All Traces And Save *)

Trace1A1Ac=Table[Tr[GS[p5].Gamma1A[[i]]. Pol[a,b] . GS[p1].Gamma1Ac[[j]]],
{i,1,6},{j,1,6}]/. kinFCI//Expand;
Trace2A2Ac=Table[Tr[GS[p3]. Gamma2A[[i]]. Pol[c,d] . GS[p2].Gamma2Ac[[j]]],
{i,1,6},{j,1,6}]/. kinFCI//Expand;

stmp=OpenWrite["Trace1A1Ac",FormatType->InputForm];
Write[stmp,Trace1A1Ac];
Close[stmp];
stmp=OpenWrite["Trace2A2Ac",FormatType->InputForm];
Write[stmp,Trace2A2Ac];
Close[stmp];

Trace1B1Bc=Table[Tr[GS[p3].Gamma1B[[i]]. Pol[a,b] . GS[p1].Gamma1Bc[[j]]],
{i,1,6},{j,1,6}]/. kinFCI//Expand;
Trace2B2Bc=Table[Tr[GS[p5].Gamma2B[[i]]. Pol[c,d] . GS[p2].Gamma2Bc[[j]]],
{i,1,6},{j,1,6}]/. kinFCI//Expand;

stmp=OpenWrite["Trace1B1Bc",FormatType->InputForm];
Write[stmp,Trace1B1Bc];
Close[stmp];
stmp=OpenWrite["Trace2B2Bc",FormatType->InputForm];
Write[stmp,Trace2B2Bc];
Close[stmp];

TraceABc=Table[Tr[GS[p5].Gamma1A[[i]]. Pol[a,b] . GS[p1].Gamma1Bc[[j]].
GS[p3].Gamma2A[[i]]. Pol[c,d] . GS[p2]. Gamma2Bc[[j]]],{i,1,6},{j,1,6}]/. kinFCI//Expand;
TraceBAc=Table[Tr[GS[p3].Gamma1B[[i]]. Pol[a,b] . GS[p1].Gamma1Ac[[j]].
GS[p5].Gamma2B[[i]]. Pol[c,d] . GS[p2]. Gamma2Ac[[j]]],{i,1,6},{j,1,6}]/. kinFCI//Expand;

stmp=OpenWrite["TraceABc",FormatType->InputForm];
Write[stmp,TraceABc];
Close[stmp];
stmp=OpenWrite["TraceBAc",FormatType->InputForm];
Write[stmp,TraceBAc];
Close[stmp];

]

```

```

(***** Program eeWene.m (Continuation) *****)
If[False,

Trace1A1Ac=<<Trace1A1Ac;
Trace2A2Ac=<<Trace2A2Ac;
Trace1B1Bc=<<Trace1B1Bc;
Trace2B2Bc=<<Trace2B2Bc;
TraceABc=<<TraceABc;
TraceBAc=<<TraceBAc;

If[True,
MsqAAc=Table[Contract[
  Trace1A1Ac[[i, j]] Trace2A2Ac[[i, j]] SumPolV[p4, mW, mu, mup]] /.
  kinFCI //Expand//Simplify, {i, 1, 6}, {j, 1, 6}];

stmp=OpenWrite["MsqAA",FormatType->InputForm];
Write[stmp,MsqAAc];
Close[stmp];

]

If[True,

FunAux=Function[{i,j}, aux1=Contract[Trace2B2Bc[[i, j]] SumPolV[p4,
mW, mu, mup]] /.
kinFCI // Expand; Contract[Trace1B1Bc[[i, j]] aux1] /. kinFCI // Expand];

MsqBBc=Table[FunAux[i,j], {i, 1, 6}, {j, 1, 6}];

stmp=OpenWrite["MsqBB",FormatType->InputForm];
Write[stmp,MsqBBc];
Close[stmp];
]

If[True,
MsqABc=Table[Contract[
  TraceABc[[i, j]] SumPolV[p4, mW, mu, mup]] /. kinFCI //Expand//Simplify, {i, 1, 6},
  {j, 1, 6}];

stmp=OpenWrite["MsqAB",FormatType->InputForm];
Write[stmp,MsqABc];
Close[stmp];

]

If[True,
MsqBAc=Table[Contract[
  TraceBAc[[i, j]] SumPolV[p4, mW, mu, mup]] /.
  kinFCI //Expand//Simplify, {i, 1, 6}, {j, 1, 6}];

stmp=OpenWrite["MsqBA",FormatType->InputForm];
Write[stmp,MsqBAc];
Close[stmp];
]

]

```



```

(***** Program eeWene.m (Continuation) *****)
If[False,

MsqAAc=<<MsqAA;
MsqBBc=<<MsqBB;
MsqABc=<<MsqAB;
MsqBAc=<<MsqBA;

Do[

If[i==j,
aux=MsqAAc[[i,i]]/DenA[[i]]/DenA[[i]]/. kinFCI//Expand//Simplify,
aux=(MsqAAc[[i,j]]+MsqAAc[[j,i]])/DenA[[i]]/DenA[[j]]/. kinFCI//Expand//Simplify];

Reaux = ComplexExpand[Re[aux]] //FCE//Simplify;

Write[stmpAll," "];
Write[stmpAll,"* AA",i,j];
Write[stmpAll,"dsig = dsig + ", Reaux /. vecmom];

Imaux=ComplexExpand[Im[aux]]//FCE//Simplify;
Write[stmpImag," "];
Write[stmpImag,"* AA",i,j];
Write[stmpImag,"dsig = dsig + ", Imaux /. vecmom];, {i,1,6}, {j,i,6}];

Write[stmpAll," "];
Write[stmpAll,"* End AA"];
Write[stmpImag," "];
Write[stmpImag,"* End AA"];

Do[

If[i==j,
aux=MsqBBc[[i,i]]/DenB[[i]]/DenBc[[i]]/. kinFCI//Expand//Simplify,
aux=(MsqBBc[[i,j]]/DenB[[i]]/DenBc[[j]]+MsqBBc[[j,i]]/DenB[[j]]/DenBc[[i]])/.
kinFCI//Expand//Simplify];

Reaux = ComplexExpand[Re[aux]] //FCE//Simplify;

Write[stmpAll," "];
Write[stmpAll,"* BB",i,j];
Write[stmpAll,"dsig = dsig + ", Reaux /. vecmom];

Imaux=ComplexExpand[Im[aux]]//FCE//Simplify;
Write[stmpImag," "];
Write[stmpImag,"* BB",i,j];
Write[stmpImag,"dsig = dsig + ", Imaux /. vecmom];, {i,1,6}, {j,i,6}];

Write[stmpAll," "];
Write[stmpAll,"* End BB"];
Write[stmpImag," "];
Write[stmpImag,"* End BB"];

```

```

(***** Program eeWene.m (Continuation) *****)
Do[
aux=(MsqABc[[i,j]]/DenBc[[j]] + MsqBAC[[j,i]]/DenB[[j]])/DenA[[i]]/. kinFCI//Expand//Simplify;
Reaux = ComplexExpand[Re[aux]]//FCE//Simplify;
Write[stmpAll," "];
Write[stmpAll,"* AB",i,j];
Write[stmpAll,"dsig = dsig + ", Reaux /. vecmom];

Imaux=ComplexExpand[Im[aux]]//FCE//Simplify;
Write[stmpImag," "];
Write[stmpImag,"* AB",i,j];
Write[stmpImag,"dsig = dsig + ", Imaux /. vecmom];, {i,1,6}, {j,1,6}];

Write[stmpAll," "];
Write[stmpAll,"* End AB"];
Write[stmpImag," "];
Write[stmpImag,"* End AB"];

]

If[Batch,Close[stmpAll]];
If[Batch,Close[stmpImag]];

Print["Time = ",TimeUsed[]-TimeNow];

```

-
- [1] R. Mertig, M. Bohm and A. Denner, Comput. Phys. Commun. **64**, 345 (1991).
[2] R. Mertig, <http://www.feyncalc.org> .
[3] J. C. Romao, <http://porthos.ist.utl.pt/CTQFT/> .
[4] T. Hahn, <http://www.feynarts.de/cuba/> .
[5] A. Pukhov, <http://theory.sinp.msu.ru/pukhov/calchep.html> .